
Harness the Power of 
Array Formulas 

 
 
 

 
 
 
 
 

Array formulas, which work on arrays of data rather than individual cells or ranges, allow users to make 
calculations that would otherwise be impossible with ordinary formulas. They can perform multiple 
discrete calculations in a single cell or return results to multiple cells. For example, an array formula can 
be used to round off totals or subtotals, a feat that is impossible to accomplish with an ordinary formula. 
Don't miss this opportunity to harness the power of array formulas to improve your analytical abilities in 
Excel. 



1 
Copyright © 2018, K2 Enterprises, LLC. Reproduction or reuse for purposes other than a K2 Enterprises’ training event is prohibited. 
 

Introduction 
Array formulas, which work on arrays of data rather than individual cells or ranges, allow users to make 
calculations that would otherwise be impossible with ordinary formulas. They can perform multiple 
discrete calculations in a single cell or return results to multiple cells. For example, an array formula can 
be used to round off totals or subtotals, a feat that is impossible to accomplish with an ordinary formula. 
Don't miss this opportunity to harness the power of array formulas to improve your analytical abilities in 
Excel. 

 
 

Learning Objectives 
Upon completing this course, participants should be able to: 

 
• Explain array formulas and how they differ from ordinary formulas 

• Identify an array formula from the Formula Bar 

• Compose array formulas from the keyboard 

• Describe common accounting situations where array formulas would be useful 

Array Formula Basics 
An array is a collection or range of two or more items. An array formula is a formula that acts on an array, 
a range of cells, instead of individual cells, or a formula that delivers results to more than one cell. In a 
nutshell, array formulas can perform multiple discrete calculations in single cell and return the results to 
a single cell or return the results to multiple cells. Note that the results of an array formula can be 
imbedded within other formulas. Since array formulas work on a group of cells in a single operation, array 
formulas can perform calculations that are not possible with conventional formulas. 

 
Excel has three types of arrays: 

 
• A reference array is an area of a worksheet that contains more than one cell. It can be an ordinary 

cell range, a worksheet reference, or a defined name. 
 

• A result array is an array of items created by an array operation. For example, the LINEST function 
can return an array of statistics than defines the results of a linear regression model. 

 

• An array constant is an array of hard-coded values (constants). An array constant can be entered 
directly into a formula or can be saved and referenced as a defined name. 

 
Array formulas can be created using ordinary functions that take individual cells as their values. This type 
of array formula must be entered using CTRL + SHIFT + ENTER. Excel also includes functions that operate 
on arrays natively, such as SUMPRODUCT, AGGREGATE, LOOKUP, INDEX and SUMIF(S), COUNTIF(S), and 
AVERAGEIF(S). These functions produce array formulas without entering CTRL + SHIFT + ENTER. Another 
set of functions produce result arrays. Among them are TRANSPOSE, TREND, FREQUENCY, LINEST, AND 
LOGEST. 

 
 

Creating Simple Array Formulas 
In this first example, three methods – conventional analysis using a helper column, an array formula using 
the SUM function, and an array formula using SUMPRODUCT – will be used to calculate the total wages 
paid to a project team for the week, as shown in Figure 1. 

 

 
Figure 1 – Team Timesheet 

1. In column D, create individual formulas to calculate the wages for each team member. Copy the 
formula down and then sum the results in cell D7. 

 
2. Build an ordinary array formula. Enter =SUM(B2:B5*C2:C5) in cell D9. Make sure to press CTRL + 

SHIFT + ENTER to enter the formula. Otherwise, the formula will evaluate to #VALUE!. Note the 
braces { } surrounding the formula in the formula bar. 

 
3. Use the SUMPRODUCT function to build the formula. Enter =SUMPRODUCT(B2:B5,C2:C5) into 

cell D10 and press ENTER. CTRL + SHIFT + ENTER is not needed to enter the formula, nor do braces 
{ } surround the formula in the formula bar. The SUMPRODUCT function produces an array 
formula without all of the complexities of an ordinary array formula. It can handle up to 255 
arrays of data. 

 

 
Figure 2 – Conventional vs Array Formulas 

 

Note that each of the three methods returned the same result, but the array formulas used a single cell 
to make the calculation. In the simple context presented, the most commonly used method with a helper 
column works as well as the array formulas, but in situations involving thousands of transactions, 
calculating the extended price or cost of goods just to calculate total revenue or total cost of goods sold 
would be cumbersome and time consuming. In those situations, a single cell array formula would be a 
better alternative. Further, using SUMPRODUCT in situations like these is a better solution because it 

calculates faster and is easier for average Excel users to understand than ordinary array formulas entered 
using CTRL + SHIFT + ENTER. 

 
Before leaving this example, let's use an advanced technique to examine or troubleshoot array formulas. 
Position the cursor in cell D9, the cell containing the ordinary array formula. Press F2 to edit the formula 
and then click number 1 in the Screen Tip that appears to select the array reference. Press F9 to see the 
underlying array, as shown in Figure 3. Do not press Enter without first pressing CTRL + Z to undo the 
change. Otherwise, the cell references will be replaced with the array constant so that the formula will no 
longer calculate by reference to the values in the cells, but by reference to the values inside the SUM 
function, thereby breaking the formula envisaged. 

 

 
Figure 3 – Troubleshooting an Array Formula 

 
Similarly, the formula built using SUMPRODUCT can be examined. Note that the individual items of the 
arrays can be examined rather than the results of the array calculations displayed in the ordinary array 
formula. 

 
 
 
 

Making Calculations Based on Conditions 
A more sophisticated example demonstrates how array formulas can be used to summarize data based 
on conditions, in this case expenses by job and by account. Three methods for making the calculation will 
be demonstrated – an array formula using the SUM function, and an array formula using SUMPRODUCT, 
and an array formula using SUMIFS. 

 
1. Build an array formula using SUM. Enter =SUM(($D$4:$D$7=$B14)*($E$4:$E$7=C$13)*($F$4:$F$7)) 

in cell C14. Note the mixed reference addressing used for cells containing the conditions so that the 
formula can be copied down and across. Make sure to press CTRL + SHIFT + ENTER to enter the 
formula. Otherwise, the formula will evaluate to #VALUE!. Note the braces { } surrounding the formula 
in the formula bar. 

F9 

Figure 5 displays the expense list and the array formula that summarizes the expenses by job and by 
account. Examine the formula's operation by using the F9 trick presented earlier. 

 
The formula first tests to see if each Account in the range D4:D7 is equal to 8000 and then tests to see if 
each Job in the range E4:E7 is equal to 701. If the result of a test is False, the formula returns 0. If the 
result is True, the formula returns 1. The results of each test are multiplied together and then multiplied 
by the requisite amount from the range F4:F7 before being summed by the formula. 

 
There is a single detail expense entry recorded for Account 8000 and Job 701, so the formula in cell C14 
evaluates to $234.00. Note that when testing multiple criteria, the multiplication operator (*) represents 
an AND condition, and the addition operator (+) represents an OR condition. 

 

 
Figure 5 - Simplified Expense Report Using Array Formula 

2. Use the SUMPRODUCT function to build the "array" formula. Position the cursor in cell C14, type in 
=SUMPRODUCT(--($D$4:$D$7=$B14),--($E$4:$E$7=C$13),$F$4:$F$7) and press ENTER. CTRL + SHIFT 
+ ENTER is not needed to enter the formula, nor do braces { } surround the formula in the formula 
bar. 

 

Use the F9 trick to examine the formula. The conditional tests add an additional layer of complexity to the 
formula requiring double negatives (--) before each conditional array. Note that the first array evaluates 
to TRUE;FALSE;FALSE;TRUE. In computing the product of the three arrays, SUMPRODUCT uses the text 
labels resulting from the conditional tests. In formulas, labels evaluate as zero, so a formula without the 
double negatives (--) in front of the conditional arrays will not calculate correctly. To force the conditional 
arrays to evaluate as 1;0;0;1, simply perform an arithmetic operation on the result array. Putting a minus 
sign in front of an array is the same as multiplying the array by -1, but that would make the tests evaluating 
to TRUE equal to -1. To overcome this limitation, place two minus signs (double negatives) in front of the 
array to convert them to positive numbers (1), which will then add and multiply correctly. 

 
3. Use the SUMIFS function to build the "array" formula. Position the cursor in cell C14, type in 

=SUMIFS($F$4:$F$7,$D$4:$D$7,$B14,$E$4:$E$7,C$13) and press ENTER. CTRL + SHIFT + ENTER is not 
needed to enter the formula, nor do braces { } surround the formula in the formula bar. 

 
Use the F9 trick to examine the formula. The conditional tests do not add any additional complexity to the 
formula because the function is designed to accommodate the conditional tests. Given the three formulas 
demonstrated, SUMIFS is clearly the best choice for making these types of conditional array calculations. 
Similarly, COUNTIFS and AVERAGEIFS can be used to perform conditional array calculations. 

 
 

Conditional Calculations other than SUM, COUNT, and AVERAGE 
Excel's functions for conditional sums, counts, and averages fit the bill in many, but not all circumstances. 
If a user wants to identify the largest, smallest, or top three observations based on specified conditions, 
Excel's built in conditional array functions can't provide a solution. Under those circumstances, users must 
create their own array formulas to make the calculations. 

 
In these examples, two formulas will be constructed – one to calculate the minimum sales by product by 
quarter using the MIN and IF functions, and one to calculate the maximum sales by product by quarter 
using the MAX and IF functions. 

 
1. Build an array formula using MIN and IF. In cell B7 enter the following formula: 

=MIN(IF(Sales!$F$2:$F$881=A7,IF(Sales!$C$2:$C$881=$E$5,Sales!$G$2:$G$881))) 

Make sure to press CTRL + SHIFT + ENTER to enter the formula. Otherwise, the formula will evaluate 
to 0. Note the braces { } surrounding the formula in the formula bar. The formula finds the minimum 
sales value that meets two conditions – whether the product name in the data record is equal to the 
product name on the report row and whether the quarter in the data record is equal to the reporting 
quarter using the IF function. Note that the value if false argument of the IF function is not specified 
for either test. A portion of the report and the formula is displayed in Figure 6. 

Figure 6 – Array Formulas to Calculate the Minimum and Maximum Values 
 

2. Similar to the formula above, build an array formula using MAX and IF. In cell C7 enter the following 
formula: 

=MAX(IF(Sales!$F$2:$F$881=A7,IF(Sales!$C$2:$C$881=$E$5,Sales!$G$2:$G$881))) 

Make sure to press CTRL + SHIFT + ENTER to enter the formula. Otherwise, the formula will evaluate 
to 0. Note the braces { } surrounding the formula in the formula bar. The formula finds the minimum 
sales value that meets two conditions – whether the product name in the data record is equal to the 
product name on the report row and whether the quarter in the data record is equal to the reporting 
quarter using the IF function. Note that the value if false argument of the IF function is not specified 
for either test. The formula is also displayed in Figure 6. 

 
Array Formulas in Reporting 
Array formulas can be used effectively in the financial reporting and analysis context. In the following 
examples, three array formulas will be examined. One will use a conventional array formula to automate 
a six-month rolling report. Another will use SUMIFs to produce a six-month summary report with user 
interaction. The third will produce a ratio analysis on an exported trial balance. 

 
Accountants often create rolling reports where a new data column is added and a prior data column is 
removed. Often, the formulas need to be recreated and copied down to complete the analysis. In this 
example, we will create a rolling report that allows a professional to copy in new data, hide the unwanted 
data, and have the analysis calculate properly without modifying any of the formulas. 

 
In cell J9 build the array formula: =SUM(($B$8:$I$8>EOMONTH(MAX($B$8:$I$8),-6))*(B9:I9)). Press CTRL 
+ SHIFT + ENTER to enter the formula. For the formula to operate correctly, the column headings in the 

report must be dates rather than labels. To add data to the report, simply insert a new column I, copy in 
the new data, and hide the oldest month. Nothing else is required. The formula automatically calculates 
the report from the most recent six months without modification. Even the date in the report heading 
updates to reflect the latest data column. The report and formula are shown in Figure 9. 

 

 
Figure 9 – Using an Array Formula to Auto-Calculate a Rolling Report 

 
The next reporting example produces a rolling six-month exception report from data stored in an Excel 
table. The table could consist of imported data or could be connected to an external data source, such as 
an accounting solution. An ODBC connection to an external data source would facilitate updating the table 
and associated reports with a single click. The formulas use the SUMIFS function to make the calculations. 
Conditional formatting is applied to the report to highlight exceptional values and end-user interaction is 
accomplished with data validation. The completed report is shown in Figure 10. 

 
Similar to the previous example, the column headings in the report must be dates rather than text labels. 
The column headings are tied by formula to the date selected by the end user in cell H4. The formula in 
the field of the report is shown below. It need not be entered using CTRL + SHIFT + ENTER because SUMIFS 
supports array calculations natively. 

 

=SUMIFS(ProductSales[Amount],ProductSales[EOM],B$5,ProductSales[Product], $A6) 

Figure 10 – Exception Report Created Using SUMIFS 
 

 
Array Constants 
Array constants are a list of values that can be used as arguments in your array formulas. 
Arrays can be either 1-dimensional or 2-dimensional depending on the number of rows and columns. 
A 1-dimensional array can exist in a single row or a single column. 
You can create an array constant and you can give it a name that can then be used in your formulas. 
 
Array constants are a list of values that can be used as arguments in your array formulas. 
Arrays can be either 1-dimensional or 2-dimensional depending on the number of rows and columns. 
A 1-dimensional array can exist in a single row or a single column. 
One dimensional array {1, 2, 3, 4} 

 

 
 
A 2-dimensional array can exist in a block of cells, made up of multiple rows and columns. 
Two dimensional array {1, 2, 3, 4; 5, 6, 7, 8; 9, 10, 11, 12} 
Notice that vertical elements are separated by a semi-colon (;) as opposed to a comma. 

An array constant can contain numbers, text, logical values as well as error values. Any text must be 
enclosed in double quotation marks (i.e. "some text"). 
An array constant cannot contain any numbers with commas, dollars, parentheses, percent signs, 
worksheet functions or any other arrays. 
Arrays do not actually need to be stored in cells and can alternatively be stored in memory during a 
calculation. 
When an array is stored in memory during a calculation the formula is executed slightly faster. 
 
One Dimensional arrays 

{"Jan", "Feb", "Mar", "Apr", "May", "Jun", "Jul", "Aug", "Sep", "Oct", "Nov", "Dec"} 

 

This array has commas separating the elements which means that the array is a horizontal array. 
You could replace the commas with semi-colon's to define a vertical array, {"Jan"; "Feb"; "Mar"; "Apr"; 
"May"; "Jun"; "Jul"; "Aug"; "Sep"; "Oct"; "Nov"; "Dec"} 
 
Two Dimensional arrays 

{"Sales","Sales","Sales","Profit","Profit","Profit";2000,2001,2002,2000,2001,2002} 

Naming an Array Constant. 

If you find yourself entering the same list again and again you can save a list as an array constant. 
Saving your list as an array constant will allow to quickly enter the list into cells. 
Select (Insert > Name > Define) to display the Define Name dialog box. 
Enter the name of your array constant and enter the array in the Refers to box. 
Notice that the array must have an equal sign before it. Without this equal sign the name will just refer to a 
text string. 

 

You must remember to enter the curly brackets as well. They will not be entered automatically in this case. 

Using the Array Constant 

Once the array constant has been defined you can enter it quickly into cells and also use it directly in your 
formulas. 
You can then enter the array constant directly in your cells {=Array_Months} 

